

# NAG Library Routine Document

## G11BBF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G11BBF computes a table from a set of classification factors using a given percentile or quantile, for example the median.

### 2 Specification

```

SUBROUTINE G11BBF (TYP, WEIGHT, N, NFAC, ISF, LFAC, IFAC, LDF, PERCNT, Y,      &
                  WT, TABLE, MAXT, NCELLS, NDIM, IDIM, ICOUNT, IWK, WK,    &
                  IFAIL)
INTEGER          N, NFAC, ISF(NFAC), LFAC(NFAC), IFAC(LDF,NFAC), LDF,      &
                MAXT, NCELLS, NDIM, IDIM(NFAC), ICOUNT(MAXT),            &
                IWK(2*NFAC+N), IFAIL
REAL (KIND=nag_wp) PERCNT, Y(N), WT(*), TABLE(MAXT), WK(2*N)
CHARACTER(1)     TYP, WEIGHT

```

### 3 Description

A dataset may include both classification variables and general variables. The classification variables, known as factors, take a small number of values known as levels. For example, the factor sex would have the levels male and female. These can be coded as 1 and 2 respectively. Given several factors, a multi-way table can be constructed such that each cell of the table represents one level from each factor. For example, the two factors sex and habitat, habitat having three levels (inner-city, suburban and rural) define the  $2 \times 3$  contingency table

Sex	Habitat		
	Inner-city	Suburban	Rural
Male			
Female			

For each cell statistics can be computed. If a third variable in the dataset was age then for each cell the median age could be computed:

Sex	Habitat		
	Inner-city	Suburban	Rural
Male	24	31	37
Female	21.5	28.5	33

That is, the median age for all observations for males living in rural areas is 37, the median being the 50% quantile. Other quantiles can also be computed: the  $p$  percent quantile or percentile,  $q_p$ , is the estimate of the value such that  $p$  percent of observations are less than  $q_p$ . This is calculated in two different ways depending on whether the tabulated variable is continuous or discrete. Let there be  $m$  values in a cell and let  $y_{(1)}, y_{(2)}, \dots, y_{(m)}$  be the values for that cell sorted into ascending order. Also, associated with each

value there is a weight,  $w_{(1)}, w_{(2)}, \dots, w_{(m)}$ , which could represent the observed frequency for that value, with  $W_j = \sum_{i=1}^j w_{(i)}$  and  $W'_j = \sum_{i=1}^j w_{(i)} - \frac{1}{2}w_{(j)}$ . For the  $p$  percentile let  $p_w = (p/100)W_m$  and  $p'_w = (p/100)W'_m$ , then the percentiles for the two cases are as given below.

If the variable is discrete, that is, it takes only a limited number of (usually integer) values, then the percentile is defined as

$$y_{(j)} \quad \text{if } W_{j-1} < p_w < W_j$$

$$\frac{y_{(j+1)} + y_{(j)}}{2} \quad \text{if } p_w = W_j.$$

If the data is continuous then the quantiles are estimated by linear interpolation.

$$y_{(1)} \quad \text{if } p'_w \leq W'_1$$

$$(1-f)y_{(j-1)} + fy_{(j)} \quad \text{if } W'_{j-1} < p'_w \leq W'_j$$

$$y_{(m)} \quad \text{if } p'_w > W'_m,$$

where  $f = (p'_w - W'_{j-1}) / (W'_j - W'_{j-1})$ .

## 4 References

John J A and Quenouille M H (1977) *Experiments: Design and Analysis* Griffin

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin

## 5 Parameters

- 1: TYP – CHARACTER(1) *Input*  
*On entry:* indicates if the variable to be tabulated is discrete or continuous.  
 TYP = 'D'  
     The percentiles are computed for a discrete variable.  
 TYP = 'C'  
     The percentiles are computed for a continuous variable using linear interpolation.  
*Constraint:* TYP = 'D' or 'C'.
- 2: WEIGHT – CHARACTER(1) *Input*  
*On entry:* indicates if there are weights associated with the variable to be tabulated.  
 WEIGHT = 'U'  
     Weights are not input and unit weights are assumed.  
 WEIGHT = 'W'  
     Weights must be supplied in WT.  
*Constraint:* WEIGHT = 'U' or 'W'.
- 3: N – INTEGER *Input*  
*On entry:* the number of observations.  
*Constraint:*  $N \geq 2$ .

- 4: NFAC – INTEGER *Input*  
*On entry:* the number of classifying factors in IFAC.  
*Constraint:*  $NFAC \geq 1$ .
- 5: ISF(NFAC) – INTEGER array *Input*  
*On entry:* indicates which factors in IFAC are to be used in the tabulation.  
 If  $ISF(i) > 0$  the  $i$ th factor in IFAC is included in the tabulation.  
 Note that if  $ISF(i) \leq 0$ , for  $i = 1, 2, \dots, NFAC$  then the statistic for the whole sample is calculated and returned in a  $1 \times 1$  table.
- 6: LFAC(NFAC) – INTEGER array *Input*  
*On entry:* the number of levels of the classifying factors in IFAC.  
*Constraint:* if  $ISF(i) > 0$ ,  $LFAC(i) \geq 2$ , for  $i = 1, 2, \dots, NFAC$ .
- 7: IFAC(LDF,NFAC) – INTEGER array *Input*  
*On entry:* the NFAC coded classification factors for the N observations.  
*Constraint:*  $1 \leq IFAC(i, j) \leq LFAC(j)$ , for  $i = 1, 2, \dots, N$  and  $j = 1, 2, \dots, NFAC$ .
- 8: LDF – INTEGER *Input*  
*On entry:* the first dimension of the array IFAC as declared in the (sub)program from which G11BBF is called.  
*Constraint:*  $LDF \geq N$ .
- 9: PERCNT – REAL (KIND=nag\_wp) *Input*  
*On entry:*  $p$ , the percentile to be tabulated.  
*Constraint:*  $0.0 < p < 100.0$ .
- 10: Y(N) – REAL (KIND=nag\_wp) array *Input*  
*On entry:* the variable to be tabulated.
- 11: WT(\*) – REAL (KIND=nag\_wp) array *Input*  
**Note:** the dimension of the array WT must be at least N if WEIGHT = 'W', and at least 1 otherwise.  
*On entry:* if WEIGHT = 'W', WT must contain the N weights. Otherwise WT is not referenced.  
*Constraint:* if WEIGHT = 'W',  $WT(i) \geq 0.0$ , for  $i = 1, 2, \dots, N$ .
- 12: TABLE(MAXT) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* the computed table. The NCELLS cells of the table are stored so that for any two factors the index relating to the factor occurring later in LFAC and IFAC changes faster. For further details see Section 8.
- 13: MAXT – INTEGER *Input*  
*On entry:* the maximum size of the table to be computed.  
*Constraint:*  $MAXT \geq$  product of the levels of the factors included in the tabulation.
- 14: NCELLS – INTEGER *Output*  
*On exit:* the number of cells in the table.

- 15: NDIM – INTEGER *Output*  
*On exit:* the number of factors defining the table.
- 16: IDIM(NFAC) – INTEGER array *Output*  
*On exit:* the first NDIM elements contain the number of levels for the factors defining the table.
- 17: ICOUNT(MAXT) – INTEGER array *Output*  
*On exit:* a table containing the number of observations contributing to each cell of the table, stored identically to TABLE.
- 18: IWK( $2 \times \text{NFAC} + \text{N}$ ) – INTEGER array *Workspace*  
 19: WK( $2 \times \text{N}$ ) – REAL (KIND=nag\_wp) array *Workspace*
- 20: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $\text{N} < 2$ ,  
 or  $\text{NFAC} < 1$ ,  
 or  $\text{LDF} < \text{N}$ ,  
 or  $\text{TYP} \neq \text{'D'}$  or  $\text{'C'}$ ,  
 or  $\text{WEIGHT} \neq \text{'U'}$  or  $\text{'W'}$ ,  
 or  $\text{PERCNT} \leq 0.0$ ,  
 or  $\text{PERCNT} \geq 100.0$ .

IFAIL = 2

On entry,  $\text{ISF}(i) > 0$  and  $\text{LFAC}(i) \leq 1$ , for some  $i$ ,  
 or  $\text{IFAC}(i, j) < 1$ , for some  $i, j$ ,  
 or  $\text{IFAC}(i, j) > \text{LFAC}(j)$ , for some  $i, j$ ,  
 or MAXT is too small,  
 or  $\text{WEIGHT} = \text{'W'}$  and  $\text{WT}(i) < 0.0$ , for some  $i$ .

IFAIL = 3

At least one cell is empty.

## 7 Accuracy

Not applicable.

## 8 Further Comments

The tables created by G11BBF and stored in TABLE and ICOUNT are stored in the following way. Let there be  $n$  factors defining the table with factor  $k$  having  $l_k$  levels, then the cell defined by the levels  $i_1, i_2, \dots, i_n$  of the factors is stored in the  $m$ th cell given by:

$$m = 1 + \sum_{k=1}^n [(i_k - 1)c_k],$$

where  $c_j = \prod_{k=j+1}^n l_k$ , for  $j = 1, 2, \dots, n-1$  and  $c_n = 1$ .

## 9 Example

The data, given by John and Quenouille (1977), is for a  $3 \times 6$  factorial experiment in 3 blocks of 18 units. The data is input in the order, blocks, factor with 3 levels, factor with 6 levels, yield, and the  $3 \times 6$  table of treatment medians for yield over blocks is computed and printed.

### 9.1 Program Text

```

Program g11bbfe

!      G11BBF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
Use nag_library, Only: g11bbf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: percent
Integer                    :: i, ifail, j, k, ldf, liwk, lwk, lwt, &
                             maxt, n, ncells, ncol, ndim, nfac,   &
                             nrow
Character (1)              :: typ, weight
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: table(:), wk(:), wt(:), y(:)
Integer, Allocatable         :: icount(:), idim(:), ifac(:, :),   &
                             isf(:), iwk(:), lfac(:)

!      .. Executable Statements ..
Write (nout,*) 'G11BBF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Read in the problem size
Read (nin,*) typ, weight, n, nfac, percent

If (weight=='W' .Or. weight=='w') Then
  lwt = n
Else
  lwt = 0
End If
liwk = 2*nfac + n
lwk = 2*n
ldf = n
Allocate (isf(nfac), lfac(nfac), ifac(ldf, nfac), idim(nfac), iwk(liwk), y(n), &
          wt(lwt), wk(lwk))

!      Read in data
If (lwt>0) Then
  Read (nin,*) (ifac(i, 1:nfac), y(i), wt(i), i=1, n)

```

```

      Else
        Read (nin,*)(ifac(i,1:nfac),y(i),i=1,n)
      End If
      Read (nin,*) lfac(1:nfac)
      Read (nin,*) isf(1:nfac)

!      Calculate the size of TABLE
      maxt = 1
      Do i = 1, nfac
        If (isf(i)>0) Then
          maxt = maxt*lfac(i)
        End If
      End Do

      Allocate (table(maxt),icount(maxt))

!      Compute classification table
      ifail = 0
      Call g11bbf(typ,weight,n,nfac,isf,lfac,ifac,ldf,percnt,y,wt,table,maxt, &
        ncells,ndim,idim,icount,iwk,wk,ifail)

!      Display results
      Write (nout,99999) ' TABLE for ', percnt, 'th percentile'
      Write (nout,*)
      ncol = idim(ndim)
      nrow = ncells/ncol
      k = 1
      Do i = 1, nrow
        Write (nout,99998)(table(j),'(',icount(j),')',j=k,k+ncol-1)
        k = k + ncol
      End Do

99999 Format (A,F4.0,A)
99998 Format (1X,6(F8.2,A,I2,A))
      End Program g11bbfe

```

## 9.2 Program Data

G11BBF Example Program Data

```

'C' 'U' 54 3 50.0
1 1 1 274
1 2 1 361
1 3 1 253
1 1 2 325
1 2 2 317
1 3 2 339
1 1 3 326
1 2 3 402
1 3 3 336
1 1 4 379
1 2 4 345
1 3 4 361
1 1 5 352
1 2 5 334
1 3 5 318
1 1 6 339
1 2 6 393
1 3 6 358
2 1 1 350
2 2 1 340
2 3 1 203
2 1 2 397
2 2 2 356
2 3 2 298
2 1 3 382
2 2 3 376
2 3 3 355
2 1 4 418
2 2 4 387
2 3 4 379

```

2 1 5 432  
 2 2 5 339  
 2 3 5 293  
 2 1 6 322  
 2 2 6 417  
 2 3 6 342  
 3 1 1 82  
 3 2 1 297  
 3 3 1 133  
 3 1 2 306  
 3 2 2 352  
 3 3 2 361  
 3 1 3 220  
 3 2 3 333  
 3 3 3 270  
 3 1 4 388  
 3 2 4 379  
 3 3 4 274  
 3 1 5 336  
 3 2 5 307  
 3 3 5 266  
 3 1 6 389  
 3 2 6 333  
 3 3 6 353  
 3 3 6  
 0 1 1

### 9.3 Program Results

G11BBF Example Program Results

TABLE for 50.th percentile

226.00( 3)	320.25( 3)	299.50( 3)	385.75( 3)	348.00( 3)	334.75( 3)
329.25( 3)	343.25( 3)	365.25( 3)	370.50( 3)	327.25( 3)	378.00( 3)
185.50( 3)	328.75( 3)	319.50( 3)	339.25( 3)	286.25( 3)	350.25( 3)

---